QT -= gui

TEMPLATE = lib
DEFINES += QTE_CORE_LIBRARY

TARGET = QTECore

include(../pri/qt-e.pri)

SOURCES += \
	qteglobal.cpp \
	log/jlog.cpp \
	log/private/jlog_p.cpp \
	corelib/jasynchronization.cpp \
	corelib/jsettings.cpp \
	corelib/jmodule.cpp \
	corelib/jsystem.cpp \
	strategy/jabstractinterfacecaller.cpp \
	strategy/jinterfacestrategy.cpp \
	strategy/jmessagenotifier.cpp

HEADERS += \
	joperate.h \
	jfunctiontraits.h \
	jtype.h \
	qteglobal.h \
	log/jlog.h \
	log/private/jlog_p.h \
	corelib/jasynchronization.h \
	corelib/jsingleton.h \
	corelib/jsettings.h \
	corelib/jmodule.h \
	corelib/jsystem.h \
	strategy/jabstractinterfacecaller.h \
	strategy/jinterfacestrategy.h \
	strategy/jmessagenotifier.h

unix {
	QT += core-private
	SOURCES += corelib/qeventdispatcher_epoll.cpp
	HEADERS += corelib/qeventdispatcher_epoll.h

	QMAKE_POST_LINK += \  # sh / bash
		mkdir -p $${includeInstallPath}/QTECore && \
		cp -fP $${proIncludePath}/QTECore/* $${includeInstallPath}/QTECore && \
		cp -fP $${sourceDir}/QTECore/*.h $${includeInstallPath}/QTECore && \
		cp -fP $${sourceDir}/QTECore/corelib/*.h $${includeInstallPath}/QTECore && \
		cp -fP $${sourceDir}/QTECore/strategy/*.h $${includeInstallPath}/QTECore && \
		cp -fP $${sourceDir}/QTECore/log/*.h $${includeInstallPath}/QTECore
}

win32 {
	QMAKE_POST_LINK += \  # dos
		$${proRootDir}\bin\mkdir.bat $${includeInstallPath}\QTECore && \
		xcopy $${proIncludePath}\QTECore\* $${includeInstallPath}\QTECore /y /a && \
		xcopy $${sourceDir}\QTECore\*.h $${includeInstallPath}\QTECore /y /a && \
		xcopy $${sourceDir}\QTECore\corelib\*.h $${includeInstallPath}\QTECore /y /a && \
		xcopy $${sourceDir}\QTECore\strategy\*.h $${includeInstallPath}\QTECore /y /a && \
		xcopy $${sourceDir}\QTECore\log\*.h $${includeInstallPath}\QTECore /y /a
}
